

Lrc: Html Combinators

Processamento de Linguagens II

LMCC & LESI, Universidade do Minho

Ano lectivo 2000/2001

João Saraiva

$\langle \text{Html.ssl} \rangle \equiv$
 $\langle \text{HTML Abstract Grammar} \rangle$
 $\langle \text{HTML Combinators} \rangle$
 $\langle \text{HTML Abstract Grammar: Pretty Printing} \rangle$

Os construtores básicos para elementos HTML são:

$\langle \text{HTML Abstract Grammar} \rangle \equiv$

```
list      Html
;
Html      : NoHtml  ()
          | ConsHtml (HtmlEl Html)
;
HtmlEl    : ConsHtmlText (STR)
          | ConsHtmlTag  (STR HtmlAttrs Html STR)
;
list      HtmlAttrs
;
HtmlAttrs : NoHtmlAttrs()
          | ConsHtmlAttrs(HtmlAttr HtmlAttrs)
;
HtmlAttr  : ConsHtmlAttr (STR STR)
;
```

De seguida apresentam-se alguns combinadores:

```

(HTML Combinators)≡
  Html text (STR s)
  { ConsHtmlText (s) :: NoHtml()
  };

  Html tag (STR s , Html h)
  { ConsHtmlTag (s,NoHtmlAttrs(),h,s) :: NoHtml()
  };

  Html tagWithAttrs (STR s, HtmlAttrs attrs, Html h)
  { ConsHtmlTag (s,attrs,h,s) :: NoHtml()
  };

  HtmlAttr attr (STR n , STR v)
  { ConsHtmlAttr (n,v)
  };

  Html font (Html h) { tag ("font",h) };

  Html bold (Html h) { tag ("b",h) };
  Html italic (Html h) { tag ("i",h) };
  Html emph (Html h) { tag ("em",h) };

  Html hline () { tag ("hr", NoHtml()) };

  Html par (Html h) { tag ("p", h) };

  Html br () { text(" ") @ tag ("br", NoHtml()) };

  Html space () { text(" ") };

  Html center (Html h) { tag ("center",h) };

  Html table (Html h)
  { tag ("table",h) };

  Html row (Html h)
  { tag ("tr",h) };

  Html rowWithAttrs (HtmlAttrs attrs,Html h)
  { tagWithAttrs("tr",attrs,h) };

  Html elem (Html h)
  { tag ("td",h) };

  Html elemWithAttrs (HtmlAttrs attrs,Html h)
  { tagWithAttrs("td",attrs,h) };

  HtmlAttr alignleft ()
  { attr ("align","left") };

  HtmlAttr alignright ()
  { attr ("align","right") };

```

```

HtmlAttr acenter ()
{ attr ("align","center") };

Html link (STR url,Html h)
{ tagWithAttrs ("a",attr("href",url)::NoHtmlAttrs(),h) };

Html image (STR url,Html h)
{ tagWithAttrs ("img",attr("src",url)::NoHtmlAttrs(),h) };

Html itemize (Html h)
{ tag ("ul",h) };

Html item (Html h)
{ tag ("li",h) };

```

$\langle \text{HTML Abstract Grammar: Pretty Printing} \rangle \equiv$

```

view      HTML
;

Html      : ConsHtml
[ HTML @ ::= @ ["%n"] @ ]
;

HtmlEl    : ConsHtmlText
[ HTML @ ::= @ ]
| ConsHtmlTag
[ HTML @ ::= "<" @ " " @ ">" @ "</" @ ">" ]
;

HtmlAttrs : ConsHtmlAttrs
[ HTML @ ::= @ [" "] @ ]
;

HtmlAttr  : ConsHtmlAttr
[ HTML @ ::= @ "=" @ "\"" @ "\""]
;

```